

Coronavirus propagation SIR model

Anton Antonov

MathematicaForPrediction at WordPress

SystemModeling at GitHub

March 2020

Introduction

The compartmental epidemiology SIR model is the simplest core model to initiate the model(s) described and worked with in the notebook "Simple Coronavirus propagation dynamics model".

The primary purpose of this notebook is to introduce both a discipline and workflow of specification, design, and experimenting with these kind of models.

Remark: We consider the coronavirus propagation models as instances of the more general System Dynamics (SD) models.

Remark: The SIR model is a classic epidemic model.

The model in this notebook differs from the classical SIR model with the following elements:

1. Two separate infected populations: one is "severely symptomatic", the other is "normally symptomatic"
2. The monetary equivalent of lost productivity due to infected or died people is tracked.

Workflow outline

The model design workflow used in this notebook has the following steps.

1. Get sufficiently familiar with the problem domain.
2. Come up with SD stocks, rates, (and loops.)
3. Write down dictionaries for the stocks and rates.
4. Write down the system of ODE's.
5. Iteratively refine, extend, or review steps 2, 3, 4.
 - 5.1. There are dedicated articles and books on just these steps.
6. Search, obtain, or otherwise come up with meaningful initial values for the stocks.
7. Search, obtain, or otherwise come up with meaningful values for the rates.
8. Determine parameters in the system of ODE's to focus on in order to investigate the behavior of the modeled system.
9. Visualize the stocks evolutions with respect to different values of the focus parameters.
 - 9.1. Use interactive interfaces for better impressions.
10. Make sensitivity analysis plots.
 - 10.1. In order to see the effects of a given parameter or a set of parameters.
11. Calibrate the model to fit observed data.
 - 11.1. This is an important step that is also optional if only qualitative results about the modeled system are of interest.

Load packages

For the interactive plots we use the functions in the package [AAp1].

```
In[445]:= Import["https://raw.githubusercontent.com/antononcube/SystemModeling/master/WL/SystemDynamicsInteractiveInterfacesFunctions.m"]
```

Model design

The description of the SIR model can be found in [Wk1, HH1]. The assumptions of the SIR model considered in this notebook can be found in [AA1].

Model dictionary

In this section we make the *model dictionary* that is used in the mathematical model formulation. We use SD’s paradigm of stocks, rates.

Stocks

```
In[446]:= aStocks = <|
  "Total Population" → TP[t],
  "Susceptible Population" → SP[t],
  "Infected Normally Symptomatic Population" → INSP[t],
  "Infected Severely Symptomatic Population" → ISSP[t],
  "Recovered Population" → RP[t],
  "Money of lost productivity" → MLP[t]
|>;
```

Rates

```
In[447]:= aRates = <|
  "Population death rate" → δ[TP],
  "Infected Normally Symptomatic Population death rate" → δ[INSP],
  "Infected Severely Symptomatic Population death rate" → δ[ISSP],
  "Severly Symptomatic Population Fraction" → sspf[SP],
  "Contact rate for the normally symptomatic population" → β[INSP],
  "Contact rate for the severely symptomatic population" → β[ISSP],
  "Average infectious period" → aip,
  "Lost productivity cost rate (per person per day)" → lpcr[ISSP, INSP]
|>;
```

Tabulation

```
In[448]:= ResourceFunction["GridTableForm"][List@@@Normal[aStocks], TableHeadings -> {"Description", "Symbol"}]
```

Out[448]=

#	Description	Symbol
1	Total Population	TP[t]
2	Susceptible Population	SP[t]
3	Infected Normally Symptomatic Population	INSP[t]
4	Infected Severely Symptomatic Population	ISSP[t]
5	Recovered Population	RP[t]
6	Money of lost productivity	MLP[t]

```
In[449]:= ResourceFunction["GridTableForm"][List@@@Normal[aRates], TableHeadings -> {"Description", "Symbol"}]
```

Out[449]=

#	Description	Symbol
1	Population death rate	$\delta[TP]$
2	Infected Normally Symptomatic Population death rate	$\delta[INSP]$
3	Infected Severely Symptomatic Population death rate	$\delta[ISSP]$
4	Severly Symptomatic Population Fraction	sspf[SP]
5	Contact rate for the normally symptomatic population	$\beta[INSP]$
6	Contact rate for the severely symptomatic population	$\beta[ISSP]$
7	Average infectious period	aip
8	Lost productivity cost rate (per person per day)	lpcr[ISSP, INSP]

Model equations

In this section we specify and explain the equations of the modified SIR model.

Equations

In this sub-section we present the equations. (The corresponding rationale / explanations are given in the next sub-section.)

In[450]:=
$$\text{newlyInfectedTerm} = \frac{\beta[ISSP]}{TP[t]} * SP[t] * ISSP[t] + \frac{\beta[INSP]}{TP[t]} * SP[t] * INSP[t];$$

In[451]:=

```
lsEquations = {
  SP'[t] == -newlyInfectedTerm - δ[TP] * SP[t],
  INSP'[t] == (1 - sspf[SP]) * newlyInfectedTerm - (1 / aip) * INSP[t] - δ[INSP] * INSP[t],
  ISSP'[t] == sspf[SP] * newlyInfectedTerm - (1 / aip) * ISSP[t] - δ[ISSP] * ISSP[t],
  RP'[t] == (1 / aip) * (ISSP[t] + INSP[t]) - δ[TP] * RP[t],
  MLP'[t] == lpcr[ISSP, INSP] * (TP[t] - RP[t] - SP[t])
};
ResourceFunction["GridTableForm"][List/@lsEquations]
```

Out[452]=

#	1
1	$SP'[t] == -\frac{INSP[t] \times SP[t] \times \beta[INSP]}{TP[t]} - \frac{ISSP[t] \times SP[t] \times \beta[ISSP]}{TP[t]} - SP[t] \times \delta[TP]$
2	$INSP'[t] == -\frac{INSP[t]}{aip} + (1 - sspf[SP]) \left(\frac{INSP[t] \times SP[t] \times \beta[INSP]}{TP[t]} + \frac{ISSP[t] \times SP[t] \times \beta[ISSP]}{TP[t]} \right) - INSP[t] \times \delta[INSP]$
3	$ISSP'[t] == -\frac{ISSP[t]}{aip} + sspf[SP] \left(\frac{INSP[t] \times SP[t] \times \beta[INSP]}{TP[t]} + \frac{ISSP[t] \times SP[t] \times \beta[ISSP]}{TP[t]} \right) - ISSP[t] \times \delta[ISSP]$
4	$RP'[t] == \frac{INSP[t] + ISSP[t]}{aip} - RP[t] \times \delta[TP]$
5	$MLP'[t] == lpcr[ISSP, INSP] (-RP[t] - SP[t] + TP[t])$

Equations explanations

Newly infected term

Since the newly infected population is going to participate in different Right Hand Sides (RHS) of the model equations we define it separately:

In[453]:=

```
newlyInfectedTerm
```

Out[453]=

$$\frac{INSP[t] \times SP[t] \times \beta[INSP]}{TP[t]} + \frac{ISSP[t] \times SP[t] \times \beta[ISSP]}{TP[t]}$$

We have two separate terms: one for the normally symptomatic infected population, the other for the severely symptomatic infected population. Each of these populations has its own contact rate, $\beta[INSP]$ and $\beta[ISSP]$ respectively.

Infected normally symptomatic population

One of the finds for the Coronavirus is that $\approx 20\%$ of the infected are severely symptomatic. This reflected in the parameter $sspf[SP]$; we multiple the newly infected by $1 - sspf[SP]$ to obtain the normally symptomatic.

The normally infected recover after average infectious period (aip) or with a certain rate $1/aip$. Note that aip the same for both normally and severely infected.

Further we subtract the number of infected that are going to die.

In[454]:=

```
INSP'[t] == (1 - sspf[SP]) * newlyInfectedTerm - (1 / aip) * INSP[t] - δ[INSP] * INSP[t]
```

Out[454]=

$$INSP'[t] == -\frac{INSP[t]}{aip} + (1 - sspf[SP]) \left(\frac{INSP[t] \times SP[t] \times \beta[INSP]}{TP[t]} + \frac{ISSP[t] \times SP[t] \times \beta[ISSP]}{TP[t]} \right) - INSP[t] \times \delta[INSP]$$

Infected severely symptomatic population

The description of this equation is analogous to the description for “Infected normally symptomatic population” equation.

Recovered population

The recovered population is obtained from:

1. The average recover period (or rate) and the normally and severely symptomatic populations
2. The number of expected to die recovered.

Money from lost productivity

We have a simple monitoring stock $MLP[t]$ that shows the amount of money lost due to people being infected or dead because of infections. We assume that both the susceptible and the recovered are productive, and infected do not work. Hence, if we subtract the susceptible and recovered from the total population and multiply the result with appropriate loss rate we can the value for $MLP[t]$.

```
In[455]:= MLP'[t] == lpcr[ISSP, INSP] * (TP[t] - RP[t] - SP[t])
Out[455]= MLP'[t] == lpcr[ISSP, INSP] (-RP[t] - SP[t] + TP[t])
```

Simulations

Parameters

The death rates and the average infectious period were obtained by World Wide Web searches. The contact rates $\beta[_]$ and the lost productivity cost rate are completely heuristic.

```
In[456]:= aRateRules = <|
  TP[t] -> 100 000,
   $\delta[TP] \rightarrow (800 / 10^5) / 365,$ 
   $\delta[ISSP] \rightarrow 0.035 / aip,$ 
   $\delta[INSP] \rightarrow 0.01 / aip,$ 
   $\beta[ISSP] \rightarrow 6,$ 
   $\beta[INSP] \rightarrow 3,$ 
  aip -> 4 * 7,
  sspf[SP] -> 0.2,
  lpcr[ISSP, INSP] -> 600
|>;
```

Equations with the actual parameters

To be used for parametric equation solving:

```
In[457]:= lsFocusParams = {aip, sspf[SP],  $\beta[ISSP]$ ,  $\beta[INSP]$ };
aParametricRateRules = KeyDrop[aRateRules, lsFocusParams];

In[459]:= lsActualEquations =
  Join[lsEquations /. Normal[aParametricRateRules],
    {
      SP[0] == (TP[t] /. aRateRules) - 2,
      ISSP[0] == 1,
      INSP[0] == 1,
      RP[0] == 0,
      MLP[0] == 0
    }
  ];
```

Solution

Here we solve a parametric system of ODE's:

```
In[460]:= aSol =  
  Association@Flatten@  
    ParametricNDSolve[lsActualEquations, {SP, ISSP, INSP, RP, MLP}, {t, 0, 365}, lsFocusParams]
```

Out[460]= \langle $\left[\begin{array}{l} \text{SP} \rightarrow \text{ParametricFunction} \left[\begin{array}{l} \text{Expression: SP} \\ \text{Parameters: \{aip, sspf[SP], \beta[ISSP], \beta[INSP]\}} \end{array} \right], \\ \\ \text{ISSP} \rightarrow \text{ParametricFunction} \left[\begin{array}{l} \text{Expression: ISSP} \\ \text{Parameters: \{aip, sspf[SP], \beta[ISSP], \beta[INSP]\}} \end{array} \right], \text{INSP} \rightarrow \text{ParametricFunction} \left[\begin{array}{l} \text{Expression: INSP} \\ \text{Parameters: \{aip, sspf[SP], \beta[ISSP], \beta[INSP]\}} \end{array} \right], \\ \\ \text{RP} \rightarrow \text{ParametricFunction} \left[\begin{array}{l} \text{Expression: RP} \\ \text{Parameters: \{aip, sspf[SP], \beta[ISSP], \beta[INSP]\}} \end{array} \right], \text{MLP} \rightarrow \text{ParametricFunction} \left[\begin{array}{l} \text{Expression: MLP} \\ \text{Parameters: \{aip, sspf[SP], \beta[ISSP], \beta[INSP]\}} \end{array} \right] \end{array} \right] \rangle$

Interactive Plots

```

In[461]:= opts = {PlotRange → All, PlotLegends → None, PlotTheme → "Detailed", PerformanceGoal → "Speed", ImageSize → 300};
lsPopulationKeys = {SP, ISSP, INSP, RP};
lsEconKeys = {MLP};
Manipulate[
  DynamicModule[{lsPopulationPlots, lsEconPlots, lsRestPlots},

    lsPopulationPlots =
      ParametricSolutionsPlots[
        aStocks,
        KeyTake[aSol, lsPopulationKeys],
        {aip, spf, crisp, criap}, ndays,
        "LogPlot" → popLogPlotQ, "Together" → popTogetherQ, "Derivatives" → popDerivativesQ, "DerivativePrefix" → "Δ", opts];

    lsEconPlots =
      ParametricSolutionsPlots[
        aStocks,
        KeyTake[aSol, lsEconKeys],
        {aip, spf, crisp, criap}, ndays,
        "LogPlot" → econLogPlotQ, "Together" → econTogetherQ, "Derivatives" → econDerivativesQ, "DerivativePrefix" → "Δ", opts];

    lsRestPlots =
      ParametricSolutionsPlots[
        aStocks,
        KeyDrop[aSol, Join[lsPopulationKeys, lsEconKeys]],
        {aip, spf, crisp, criap}, ndays,
        "LogPlot" → econLogPlotQ, "Together" → econTogetherQ, "Derivatives" → econDerivativesQ, "DerivativePrefix" → "Δ", opts];

    Multicolumn[Join[lsPopulationPlots, lsEconPlots, lsRestPlots], nPlotColumns, Dividers → All, FrameStyle → GrayLevel[0.8]]
  ],
  {{aip, 21., "Average infectious period (days)"}, 1, 60., 1, Appearance → {"Open"}},
  {{spf, 0.2, "Severely symptomatic population fraction"}, 0, 1, 0.025, Appearance → {"Open"}},
  {{crisp, 6, "Contact rate of the infected severely symptomatic population"}, 0, 30, 0.1, Appearance → {"Open"}},
  {{criap, 3, "Contact rate of the infected normally symptomatic population"}, 0, 30, 0.1, Appearance → {"Open"}},
  {{ndays, 90, "Number of days"}, 1, 365, 1, Appearance → {"Open"}},
  {{popTogetherQ, True, "Plot populations together"}, {False, True}},
  {{popDerivativesQ, False, "Plot populations derivatives"}, {False, True}},
  {{popLogPlotQ, False, "LogPlot populations"}, {False, True}},
  {{econTogetherQ, False, "Plot economics functions together"}, {False, True}},
  {{econDerivativesQ, False, "Plot economics functions derivatives"}, {False, True}},
  {{econLogPlotQ, False, "LogPlot economics functions"}, {False, True}},
  {{nPlotColumns, 2, "Number of plot columns"}, Range[5]},
  ControlPlacement → Left, ContinuousAction → False]

```

