

Epidemiology Compartmental Modeling Monad Demo

Version 0.4

Anton Antonov
MathematicaForPrediction at WordPress
SystemModeling at GitHub
April 2020

Introduction

This preliminary notebook aims at giving short, actionable overview of a software monad for rapid specification of Epidemiology Compartmental Modeling (ECM) workflows. Monad’s name is “ECMMon”, which stands for “**E**pidemiol-
ogy **C**ompartmental **M**odeling **M**onad”.

The software monad ECMMon is implemented in [AAp6], which relies on [AAp1-AAp4]. The original workflow discussed in [AA4] was implemented in [AAp5].

We use real-life data, The New York Times COVID-19 data, see [NYT1, AA5].

Load packages

```
In[75]:= Import["https://raw.githubusercontent.com/antononcube/SystemModeling/master/Projects/Coronavirus-propagation-dynamics/WL/MonadicEpidemiologyCompartmentalModeling.m"];
Import["https://raw.githubusercontent.com/antononcube/SystemModeling/master/Projects/Coronavirus-propagation-dynamics/WL/MultiSiteModelSimulation.m"]

» Importing from GitHub: HextileBins.m
» Importing from GitHub: EpidemiologyModels.m
» Importing from GitHub: EpidemiologyModelModifications.m
» Importing from GitHub: EpidemiologyModelingSimulationFunctions.m
» Importing from GitHub: EpidemiologyModelingSimulationFunctions.m
» Importing from GitHub: SystemDynamicsInteractiveInterfacesFunctions.m
```


Ingest data

General design

Flow-chart

The following flow-chart visualizes the workflows given below:

```
Import["https://github.com/antononcube/SystemModeling/raw/master/Projects/Coronavirus-propagation-dynamics/ECMMon-workflow.jpeg"]
```



Two models in the monad

- An ECMMon object can have one or two models. One of the models is a “seed”, single-site model from [AAp1], which, if desired, is scaled into a multi-site model, [AA3, AAp2].
- Workflows with only the single-site model are supported.
 - Say, workflows for doing sensitivity analysis, [AA6, BC1].

- Scaling of a single-site model into multi-site is supported and facilitated.
- Workflows for the multi-site model include preliminary model scaling steps and simulation steps.
- After the single-site model is scaled the monad functions use the multi-site model.
- The workflows should be easy to specify and read.

Single-site model workflow

1. Make a single-site model.
2. Assign stocks initial conditions.
3. Assign rates values.
4. Simulate.
5. Plot results.
6. Go to 2.

Multi-site model workflow

1. Make a single-site model.
2. Assign initial conditions and rates.
3. Scale the single-site model into a multi-site model.
 - 3.1. The single-site assigned rates become “global” when the single-site model is scaled.
 - 3.2. The scaling is based on assumptions for traveling patterns of the populations.
 - 3.3. There are few alternatives for that scaling:
 - 3.3.1. Using locations geo-coordinates
 - 3.3.2. Using regular grids covering a certain area based on in-habited locations geo-coordinates
 - 3.3.3. Using traveling patterns contingency matrices
 - 3.3.4. Using “artificial” patterns of certain regular types for qualitative analysis purposes
4. Enhance the multi-site traveling patterns matrix and re-scale the single site model.
 - 4.1. We might want to combine traveling patterns by ground transportation with traveling patterns by airplanes.
 - 4.2. For quarantine scenarios this might a less important capability of the monad.
 - 4.2.1. Hence, this an optional step.
5. Assign stocks initial conditions for each of the sites in multi-scale model.
6. Assign rates for each of the sites.
7. Simulate.
8. Plot global simulation results.
9. Plot simulation results for focus sites.

Single-site workflow demo

In this section we demonstrate some of the sensitivity analysis discussed in [AA6, BC1].

Make a single-site model:


```
In[84]:= model1 = SEI2HREconModel[t, "InitialConditions" → True, "RateRules" → True, "TotalPopulationRepresentation" → "AlgebraicEquation"];
```

Execute the workflow multiple times with different quarantine starts:

```
In[85]:= qlVar = 56;
lsRes =
Map[
  ECMMonUnit[] ⇒
    ECMMonSetSingleSiteModel[model1] ⇒
    ECMMonAssignRateRules[
      <|
        aip → 26,
        aincp → 5,
        β[ISSP] → 0.5 * Piecewise[{{1, t < qsd}, {qcrf, qsd ≤ t ≤ qsd + ql}}, 1],
        β[INSP] → 0.5 * Piecewise[{{1, t < qsd}, {qcrf, qsd ≤ t ≤ qsd + ql}}, 1],
        qsd → #,
        ql → qlVar,
        qcrf → 0.25,
        β[HP] → 0.01,
        μ[ISSP] → 0.035 / aip,
        μ[INSP] → 0.01 / aip,
        nhbr[TP] → 3 / 1000,
        lpcr[ISSP, INSP] → 1,
        hscr[ISSP, INSP] → 1
      |>
    ] ⇒
    ECMMonSimulate[365] ⇒
    ECMMonPlotSolutions[{"Infected Severely Symptomatic Population"}, 240, "Together" → True, "Derivatives" → False,
      PlotRange → {0, 12 000}, ImageSize → 250,
      Epilog → {Orange, Dashed, Line[{{#, 0}, {#, 12 000}}], Line[{{# + qlVar, 0}, {# + qlVar, 12 000}}]},
      PlotLabel → Row[{"Quarantine start:", Spacer[5], #, ",", Spacer[5], "end:", Spacer[5], # + qlVar}],
      "Echo" → False] ⇒
    ECMMonTakeValue &, Range[25, 100, 5]];
```

Plot the simulation solutions for “Infected Severely Symptomatic Population”:

In[87]:= Multicolumn[#, 1] & /@ lsRes, 4]



Multi-site workflow demo

Here we specify a multi-site model workflow over USA data, [WRI2, NYT1].

In[88]:= model1 = SEI2HRModel[t, "InitialConditions" → True, "RateRules" → True, "TotalPopulationRepresentation" → "AlgebraicEquation"];

```

In[89]:= ecmObj =
  ECMMonUnit[] =>
    ECMMonSetSingleSiteModel[model1] =>
      ECMMonAssignRateRules[
        <|
          aip -> 26,
          aincp -> 5,
           $\beta$ [ISSP] -> 0.5 * Piecewise[{{1, t < qsd}, {qcrf, qsd ≤ t ≤ qsd + ql}}, 1],
           $\beta$ [INSP] -> 0.5 * Piecewise[{{1, t < qsd}, {qcrf, qsd ≤ t ≤ qsd + ql}}, 1],
          qsd -> 0,
          ql -> 56,
          qcrf -> 0.25,
           $\beta$ [HP] -> 0.01,
           $\mu$ [ISSP] -> 0.035 / aip,
           $\mu$ [INSP] -> 0.01 / aip,
          nhbr[TP] -> 3 / 1000,
          lpcr[ISSP, INSP] -> 1,
          hscr[ISSP, INSP] -> 1
        |>
      ] =>
    ECMMonEcho[Style["Show the single-site model tabulated form:", Bold, Purple]] =>
    ECMMonEchoFunctionContext[Magnify[ModelGridTableForm[#singleSiteModel], 0.6] &] =>
    ECMMonMakeHexagonalGrid[Keys[aPopulations], 2.5] =>
    ECMMonEcho[Style["Show the grid based on population coordinates:", Bold, Purple]] =>
    ECMMonPlotGrid["CellIDs" -> True, ImageSize -> Large] =>
    ECMMonExtendByGrid[aPopulations, 0.12] =>
    ECMMonAssignInitialConditions[aPopulations, "Total Population", "Default" -> 0] =>
    ECMMonAssignInitialConditions[DeriveSusceptiblePopulation[aPopulations, aInfected, aDead], "Susceptible Population", "Default" -> 0] =>
    ECMMonAssignInitialConditions[<||>, "Exposed Population", "Default" -> 0] =>
    ECMMonAssignInitialConditions[aInfected, "Infected Normally Symptomatic Population", "Default" -> 0] =>
    ECMMonAssignInitialConditions[<||>, "Infected Severely Symptomatic Population", "Default" -> 0] =>
    ECMMonEcho[Style["Show total populations initial conditions data:", Bold, Purple]] =>
    ECMMonPlotGridHistogram[aPopulations, ImageSize -> Large, PlotLabel -> "Total populations"] =>
    ECMMonEcho[Style["Show infected and deceased initial conditions data:", Bold, Purple]] =>
    ECMMonPlotGridHistogram[aInfected, ColorFunction -> ColorData["RoseColors"], "ShowDataPoints" -> False, ImageSize -> Large, PlotLabel -> "Infected"] =>
    ECMMonPlotGridHistogram[aDead, ColorFunction -> ColorData["RoseColors"], "ShowDataPoints" -> False, ImageSize -> Large, PlotLabel -> "Deceased"] =>
    ECMMonEcho[Style["Simulate:", Bold, Purple]] =>
    ECMMonSimulate[365] =>
    ECMMonEcho[Style["Show global population simulation results:", Bold, Purple]] =>
    ECMMonPlotSolutions[__ ~~ "Population", 365] =>
    ECMMonEcho[Style["Show site simulation results for Miami and New York areas:", Bold, Purple]] =>
    ECMMonPlotSiteSolutions[{160, 174}, __ ~~ "Population", 365] =>
    ECMMonEcho[Style["Show deceased and hospitalized populations results for Miami and New York areas:", Bold, Purple]] =>
    ECMMonPlotSiteSolutions[{160, 174}, {"Deceased Infected Population", "Hospitalized Population", "Hospital Beds"}, 300, "FocusTime" -> 120];

» Show the single-site model tabulated form:

```

»

(Stocks →

#	Symbol	Description
1	TP[t]	Total Population
2	SP[t]	Susceptible Population
3	EP[t]	Exposed Population
4	INSP[t]	Infected Normally Symptomatic Population
5	ISSP[t]	Infected Severely Symptomatic Population
6	RP[t]	Recovered Population
7	MLP[t]	Money of Lost Productivity
8	HP[t]	Hospitalized Population
9	DIP[t]	Deceased Infected Population
10	HB[t]	Hospital Beds
11	MHS[t]	Money for Hospital Services

, Rates →

#	Symbol	Description
1	μ [TP]	Population death rate
2	μ [INSP]	Infected Normally Symptomatic Population death rate
3	μ [ISSP]	Infected Severely Symptomatic Population death rate
4	sspf[SP]	Severely Symptomatic Population Fraction
5	β [INSP]	Contact rate for the normally symptomatic population
6	β [ISSP]	Contact rate for the severely symptomatic population
7	aip	Average infectious period
8	aincp	Average incubation period
9	lpcr[ISSP, INSP]	Lost productivity cost rate (per person per day)
10	μ [HP]	Hospitalized Population death rate
11	β [HP]	Contact rate for the hospitalized population
12	nhbr[TP]	Number of hospital beds rate
13	hscr[ISSP, INSP]	Hospital services cost rate (per bed per day)
14	nhbcr[ISSP, INSP]	Number of hospital beds change rate (per day)

,

Equations →

#	Equation
1	$TP'[t] = \text{Max}[0, EP[t] + INSP[t] + ISSP[t] + RP[t] + SP[t]]$
2	$SP'[t] = -\frac{HP[t] SP[t] \beta[HP]}{TP[t]} - \frac{INSP[t] SP[t] \beta[INSP]}{TP[t]} - \frac{\text{Max}[0, -HP[t] - ISSP[t]] SP[t] \beta[ISSP]}{TP[t]} - SP[t] \mu[TP]$
3	$EP'[t] = \frac{HP[t] SP[t] \beta[HP]}{TP[t]} + \frac{INSP[t] SP[t] \beta[INSP]}{TP[t]} + \frac{\text{Max}[0, -HP[t] - ISSP[t]] SP[t] \beta[ISSP]}{TP[t]} - EP[t] \left(\frac{1}{aincp} + \mu[TP] \right)$
4	$INSP'[t] = -\frac{INSP[t]}{aip} + \frac{EP[t] (1 - sspf[SP])}{aincp} - INSP[t] \mu[INSP]$
5	$ISSP'[t] = -\frac{ISSP[t]}{aip} + \frac{EP[t] sspf[SP]}{aincp} - HP[t] \mu[HP] - (-HP[t] + ISSP[t]) \mu[ISSP]$
6	$HP'[t] = \left(\begin{array}{l} \text{Min}[HB[t] - HP[t], \frac{EP[t] sspf[SP]}{aincp}] \text{ if } HP[t] < HB[t] \\ 0 \text{ if } \text{True} \end{array} \right) - \frac{HP[t]}{aip} - HP[t] \mu[HP]$
7	$RP'[t] = \frac{INSP[t] - ISSP[t]}{aip} - RP[t] \mu[TP]$
8	$DIP'[t] = HP[t] \mu[HP] + INSP[t] \mu[INSP] + (-HP[t] + ISSP[t]) \mu[ISSP]$
9	$HB'[t] = HB[t] nhbr[TP]$
10	$MHS'[t] = HP[t] hscr[ISSP, INSP]$
11	$MLP'[t] = lpcr[ISSP, INSP] (INSP[t] + ISSP[t] + HP[t] \mu[HP] + INSP[t] \mu[INSP] + (-HP[t] + ISSP[t]) \mu[ISSP])$

, RateRules →

#	Symbol	Value
1	μ [TP]	$\frac{1}{45625}$
2	μ [ISSP]	$\frac{0.035}{aip}$
3	μ [INSP]	$\frac{0.01}{aip}$
4	β [ISSP]	$0.5 \left(\begin{array}{l} 1 \text{ if } t < qsd \\ qcrf \text{ if } qsd \leq t \leq ql + qsd \\ 1 \text{ if } \text{True} \end{array} \right)$
5	β [INSP]	$0.5 \left(\begin{array}{l} 1 \text{ if } t < qsd \\ qcrf \text{ if } qsd \leq t \leq ql + qsd \\ 1 \text{ if } \text{True} \end{array} \right)$
6	aip	26
7	aincp	5
8	sspf[SP]	0.2
9	lpcr[ISSP, INSP]	1
10	μ [HP]	$0.25 \mu[ISSP]$
11	β [HP]	0.01
12	nhbr[TP]	$\frac{3}{1000}$
13	nhbcr[ISSP, INSP]	0
14	hscr[ISSP, INSP]	1
15	qsd	0
16	ql	56
17	qcrf	0.25


, InitialConditions →

#	Equation
1	$SP[0] = 99998$
2	$EP[0] = 0$
3	$ISSP[0] = 1$
4	$INSP[0] = 1$
5	$RP[0] = 0$
6	$MLP[0] = 0$
7	$TP[0] = 100000$
8	$HP[0] = 0$
9	$DIP[0] = 0$
10	$HB[0] = nhbr[TP] TP[0]$
11	$MHS[0] = 0$


)

» Show the grid based on population coordinates:



» grid:




» Show total populations initial conditions data:




» Show infected and deceased initial conditions data:




» Simulate:

 **NDSolve:** NDSolve has computed initial values that give a zero residual for the differential–algebraic system, but some components are different from those specified. If you need them to be satisfied, giving initial conditions for all dependent variables and their derivatives is recommended.


» Show global population simulation results:



» Show site simulation results for Miami and New York areas:



» Show deceased and hospitalized populations results for Miami and New York areas:



References

Articles

[Wk1] Wikipedia entry, "Compartmental models in epidemiology".

[Wl2] Wikipedia entry, "Coronavirus disease 2019".

[HH1] Herbert W. Hethcote (2000). "The Mathematics of Infectious Diseases". SIAM Review. 42 (4): 599–653. Bibcode:2000SIAMR..42..599H. doi:10.1137/s0036144500371907.

[BC1] Lucia Breierova, Mark Choudhari, An Introduction to Sensitivity Analysis, (1996), Massachusetts Institute of Technology.

[AA1] Anton Antonov, "Coronavirus propagation modeling considerations", (2020), SystemModeling at GitHub.

[AA2] Anton Antonov, "Basic experiments workflow for simple epidemiological models", (2020), SystemModeling at GitHub.

[AA3] Anton Antonov, "Scaling of Epidemiology Models with Multi-site Compartments", (2020), SystemModeling at GitHub.

[AA4] Anton Antonov, "WirVsVirus hackathon multi-site SEI2R over a hexagonal grid graph", (2020), SystemModeling at GitHub.

[AA5] Anton Antonov, "NY Times COVID-19 data visualization", (2020), SystemModeling at GitHub.

[AA6] Anton Antonov, "SEI2HR model with quarantine scenarios", (2020), SystemModeling at GitHub.

[AA7] Anton Antonov, "SEI2HR-Econ model with quarantine and supplies scenarios", (2020), SystemModeling at GitHub.

Repositories, packages

[WRI1] Wolfram Research, Inc., "Epidemic Data for Novel Coronavirus COVID-19", WolframCloud.

[WRI2] Wolfram Research Inc., USA county records, (2020), System Modeling at GitHub.

[NYT1] The New York Times, Coronavirus (Covid-19) Data in the United States, (2020), GitHub.

[AAr1] Anton Antonov, Coronavirus propagation dynamics project, (2020), SystemModeling at GitHub.

[AAp1] Anton Antonov, "Epidemiology models Mathematica package", (2020), SystemModeling at GitHub.

[AAp2] Anton Antonov, "Epidemiology models modifications Mathematica package", (2020), SystemModeling at GitHub.

[AAp3] Anton Antonov, "Epidemiology modeling visualization functions Mathematica package", (2020), SystemModeling at GitHub.

[AAp4] Anton Antonov, "System dynamics interactive interfaces functions Mathematica package", (2020), SystemModeling at GitHub.

[AAp5] Anton Antonov, "Multi-site model simulation Mathematica package", (2020), SystemModeling at GitHub.

[AAp6] Anton Antonov, "Monadic Epidemiology Compartmental Modeling Mathematica package", (2020), SystemModeling at GitHub.